


PRÁCTICA 3: Cálculo Integral de funciones de una variable

Cálculo de integrales

La principal orden de *Maxima* para calcular integrales es `integrate`. Nos va a permitir calcular integrales, tanto definidas como indefinidas, con mucha comodidad.

<code>integrate(f(x), x)</code>	primitiva de la función $f(x)$
<code>integrate(f(x), x, a, b)</code>	$\int_a^b f(x) dx$

Disponemos también de la siguiente opción en el menú para calcular integrales, sin necesidad de escribir el comando correspondiente en la ventana de entradas: **Análisis** → **Integrar**. Después sólo tenemos que rellenar los datos que nos interesen.



Comencemos por integrales indefinidas. *Maxima* calcula primitivas de funciones trigonométricas,

(%i1) `integrate(sin(x)*x,x);`

(%o1) $\sin(x) - x \cos(x)$

de funciones racionales,

(%i2) `integrate(1/(x^4-1),x);`

(%o2) $-\frac{\log(x+1)}{4} - \frac{\operatorname{atan}(x)}{2} + \frac{\log(x-1)}{4}$

irracionales,

(%i3) `integrate(sqrt(x^2+1),x);`

(%o3) $\frac{\operatorname{asinh}(x)}{2} + x \sqrt{x^2+1}$

ya sea aplicando integración por partes

Práctica 3: Cálculo Integral de Funciones de una variable

(%i4) `integrate(x^3*e^x,x);`

(%o4) $(x^3 - 3x^2 + 6x - 6)e^x$

o el método que considere necesario

(%i5) `integrate(%e^(-x^2),x);`

(%o5) $\frac{\sqrt{\pi} \operatorname{erf}(x)}{2}$

integral que no sabíamos hacer. Bueno, como puedes ver, *Maxima* se defiende bien con integrales.

Para calcular integrales definidas sólo tenemos que añadir los extremos del intervalo de integración

(%i6) `integrate(sin(x)*x,x,%pi/2,%pi);`

(%o6) $\pi - 1$

(%i7) `integrate(1/(x^4-1),x,0,1/2);`

(%o7) $-\frac{\log(2)}{4} - \frac{\log\left(\frac{3}{2}\right)}{4} - \frac{\operatorname{atan}\left(\frac{1}{2}\right)}{2}$

En las integrales pueden aparecer parámetros y si *Maxima* tiene “dudas” acerca de su valor pregunta,

(%i8) `integrate(x^n,x);`

Is n equal to -1?n;

(%o8) $\frac{x^{n+1}}{n+1}$

(%i9) `integrate(x^n,x);`

Is n equal to -1?y;

(%o9) $\log(x)$

Evidentemente hay funciones a las que *Maxima* no sabe calcular una integral indefinida. En ese caso, da como respuesta la misma integral que le hemos preguntado.

(%i10) `integrate(%e^(x^3+x),x);`

(%o10) $\int e^{x^3+x} dx$

Veremos que este problema desaparece cuando pasamos a integrales definidas.

Integración impropia

Como recordarás de las clases teóricas, la integral que, en principio, se define para funciones continuas en intervalos compactos, puede extenderse a funciones continuas definidas en intervalos de longitud infinita y a funciones que no están acotadas en un intervalo de longitud finita. Es lo que se conoce como *integración impropia*. Para el programa *Maxima*, trabajar con integrales impropias no supone ningún problema ya que las trata exactamente igual que las integrales de funciones continuas en intervalos cerrados y acotados. De hecho, no hay forma de distinguir unas de otras: en ningún momento hemos indicado al comando `integrate` que la función era acotada ni hemos dicho si el intervalo era abierto, cerrado o semiabierto. Sólo hemos indicado los extremos y ya está. Por tanto, con la misma orden `integrate` se pueden calcular integrales impropias. Por ejemplo:

```
(%i11) integrate(1/sqrt(1-x^2),x,-1,1);
```

```
(%o11)  $\pi$ 
```

```
(%i12) integrate(%e^(-x^2),x,0,inf);
```

```
(%o12)  $\frac{\sqrt{\pi}}{2}$ 
```

```
(%i13) integrate(1/x^2,x,1,inf);
```

```
(%o13) 1
```

```
(%i14) integrate(1/x,x,1,inf);
```

```
defint: integral is divergent.
```

```
-- an error. To debug this try: debugmode(true);
```

Observa cómo en la salida de la última integral el programa advierte de la “no convergencia” de la integral planteada. ¿Sabéis decir el por qué? Bueno, recordemos que la definición de dicha integral es

$$\int_1^{+\infty} \frac{1}{x} dx = \lim_{x \rightarrow +\infty} G(x) - \lim_{x \rightarrow 1} G(x),$$

donde G es una primitiva de $\frac{1}{x}$. En este caso, una primitiva es $\ln(x)$ que no tiene límite en $+\infty$ (es $+\infty$).

Integración numérica

Vamos a calcular de forma aproximada integrales definidas. Para ello disponemos de los siguientes comandos:

<code>quad_quags(f(x), x, a, b)</code>	aproximación numérica de $\int_a^b f(x) dx$
<code>romberg(f(x), x, a, b)</code>	aproximación numérica de $\int_a^b f(x) dx$

Práctica 3: Cálculo Integral de Funciones de una variable

En la práctica, no siempre es fácil calcular una primitiva, pero sí es fácil (para un ordenador y teóricamente) aproximar el valor de la integral por las áreas de los rectángulos que aparecen en la definición de integral. Este método no es el mejor, pero versiones mejoradas permiten aproximar el valor de casi cualquier integral.

Si en el menú **Análisis**→**Integrar** marcamos “Integración definida” e “Integración numérica” se nos da la opción de escoger entre dos métodos: quadpack y romberg, cada uno referido a un tipo diferente de aproximación numérica de la integral. Vamos a calcular numéricamente la integral que hemos hecho más arriba que tenía como resultado $\pi - 1$

```
(%i15) quad_qags(sin(x)*x,x,%pi/2,%pi);
```

```
(%o15) [2.141592653589794,2.377645473383477 10-14,21,0]
```

```
(%i16) romberg(sin(x)*x,x,%pi/2,%pi);
```

```
(%o16) 2.141591640806944
```

Observamos que hay una diferencia entre las salidas de ambos comandos. Mientras que en la última aparece el valor aproximado de la integral (fíjate que es $\pi - 1$), en la primera aparece como salida una lista de 4 valores:

- la aproximación de la integral
- el error absoluto estimado de la aproximación
- el número de evaluaciones del integrando
- el código de error (que puede ir desde 0 hasta 6). Ver ayuda del *Maxima*

Qué duda cabe que este tipo de integración numérica es interesante para integrandos de los cuales no se conoce una primitiva. Vamos a intentar calcular una primitiva y posteriormente una integral definida de la función $f(x) = e^{-x^2}$.

```
(%i17) integrate(%e-x^2,x);
```

```
(%o17)  $\frac{\sqrt{\pi} \operatorname{erf}(x)}{2}$ 
```

```
(%i18) integrate(%e-x^2,x,0,1);
```

```
(%o18)  $\frac{\sqrt{\pi} \operatorname{erf}(1)}{2}$ 
```

Si ahora queremos, podemos calcular numéricamente este valor,

```
(%i19) float(%);
```

```
(%o19) 0.7468241328124269
```

o también, podíamos haber utilizado los comandos de integración numérica directamente

```
(%i20) quad_qags(%e-x^2,x,0,1);
```

```
(%o20) [0.7468241328124271,8.291413475940727 10-15,21,0]
```

```
(%i21) romberg(%e-x^2,x,0,1);
```

```
(%o21) 0.7468241699098985
```

Aplicaciones

Cálculo de áreas planas


Te recuerdo que si f y g son funciones integrables definidas en un intervalo $[a, b]$, el área entre las dos funciones es

$$\int_a^b |f(x) - g(x)| dx$$

Evidentemente, calcular una integral con un valor absoluto no es fácil (¿cuál es la primitiva de la función $|x|$?). Como función definida a trozos, lo que tenemos que hacer es dividir el intervalo $[a, b]$ en subintervalos en los que sepamos cuál de las dos funciones f y g es la más grande.

Ejemplo Calculemos el área entre las curvas $y = x^2$, $y^2 = 8x$. Podemos dibujar las dos curvas y hacernos una idea del aspecto del área que queremos calcular.

```
(%i22) wxdraw2d(color=red,explicit(x^2,x,0,3),
color=blue,implicit(y^2-8*x,x,0,3,y,-4,4));
```



En realidad, lo primero que nos hace falta averiguar son los puntos de corte:

```
(%i23) solve([y=x^2,y^2=8*x],[x,y]);
```

```
(%o23) [[x=2,y=4],[x=-sqrt(3)*i-1,y=2*sqrt(3)*i-2],[x=sqrt(3)*i-1,y=-2*sqrt(3)*i-2],
[x=0,y=0]]
```

De todas las soluciones, nos quedamos con las soluciones reales: $(0, 0)$ y $(2, 4)$. En el intervalo $[0, 2]$, ¿cuál de las dos funciones es mayor? Son dos funciones continuas en un intervalo que sólo coinciden en 0 y en 2, por tanto el Teorema de los ceros de Bolzano nos dice que una de ellas es siempre mayor que la otra. Para averiguarlo sólo tenemos que evaluar en algún punto entre 0 y 2. En este caso es más fácil: se ve claramente que $y = x^2$ está por debajo. Por tanto el área es

```
(%i24) integrate(sqrt(8*x)-x^2,x,0,2);
```

```
(%o24) 8/3
```


Observación La fórmula que define el cálculo del área entre dos funciones tiene un valor absoluto que hace difícil calcular una primitiva directamente. Ese es el motivo por el que dividimos el intervalo en trozos: quitar ese valor absoluto. Si sólo pretendemos calcular el valor numéricamente, entonces el valor absoluto no es un impedimento y podemos calcular directamente la integral olvidándonos de puntos de corte o de qué función es mayor:

Práctica 3: Cálculo Integral de Funciones de una variable

```
(%i25) quad_qags(abs(x^2-sqrt(8*x)),x,0,2);
```

```
(%o25) [2.666666666666667,2.960594732333752 10-15,231,0]
```

Prueba esto que hemos comentado con el siguiente ejemplo:



Nota: Otro aspecto muy a tener en cuenta es cómo obtener los puntos de corte... no siempre el comando **solve** podrá calcularlos. ¿Te acuerdas de los otros comandos que vimos?

Prueba con el siguiente ejemplo: Calcular el área encerrada entre estas dos curvas en el intervalo indicado:

$$y = \sin x, y = \sin 2x, x = 0, x = \pi/2$$

Longitud de curvas

Si f es una función definida en el intervalo $[a, b]$, la longitud del arco de curva entre $(a, f(a))$ y $(b, f(b))$ se puede calcular mediante la fórmula

$$\text{longitud} = \int_a^b \sqrt{1 + f'(x)^2} dx$$

Por ejemplo, la semicircunferencia superior de radio 1 centrada en el origen es la gráfica de la función $f(x) = \sqrt{1 - x^2}$, con x variando entre -1 y 1. Si aplicamos la fórmula anterior podemos calcular la longitud de una circunferencia.

```
(%i26) f(x):=sqrt(1-x^2)
```

```
(%i27) diff(f(x),x);
```

```
(%o27) - 
$$\frac{x}{\sqrt{1-x^2}}$$


```

```
(%i28) integrate(sqrt(1+diff(f(x),x)^2),x,-1,1);
```

```
(%o28)  $\pi$ 
```

Observa que hemos calculado la longitud de media circunferencia, ya que la longitud de la circunferencia completa de radio 1 es 2π .

```
(%i30) wxplot2d([f(x)], [x,-1,1],same_xy)$
```



Volúmenes de revolución

Giro de una curva $y = f(x)$ respecto al eje X

Sea $f : [a, b] \rightarrow \mathbb{R}$ una función continua. Girando la región del plano comprendida entre la curva $y = f(x)$, el eje de abscisas con x entre a y b , alrededor del eje OX obtenemos un sólido de revolución Ω con volumen igual a

$$\text{Vol}(\Omega) = \pi \int_a^b f(x)^2 dx$$

Ejemplo Calculemos el volumen de una esfera de radio R viéndola como superficie de revolución. La curva que consideramos es $f(x) = \sqrt{R^2 - x^2}$ con $x \in [-R, R]$ y, por tanto, el volumen lo calculamos aplicando la fórmula anterior :

```
(%i33) %pi*integrate(R^2-x^2,x,-R,R);
```

```
(%o33)  $\frac{4 \pi R^3}{3}$ 
```

Área de superficies de revolución

Igual que hemos visto cómo podemos calcular el volumen de una figura obtenida girando una función respecto a alguno de los ejes, también podemos calcular el área de la superficie Ω obtenida al girar respecto al eje OX una función f . El área al girar f en el intervalo $[a, b]$ es

$$\text{área}(\Omega) = 2\pi \int_a^b f(x) \sqrt{1 + f'(x)^2} dx$$

Por ejemplo, una esfera de radio 1 la podemos obtener girando respecto del eje OX la función $\sqrt{1 - x^2}$.

```
(%i35) f(x):=sqrt(1-x^2)$
```

```
(%i36) 2*%pi*integrate(f(x)*sqrt(1+diff(f(x),x)^2),x,-1,1);
```

```
(%o36) 4 pi
```

EJERCICIOS PROPUESTOS

1) Calcular una primitiva de las siguientes funciones:

$$a) f(x) = \cos^5 x$$

$$b) f(x) = \frac{1}{1+x^4}$$

$$c) f(x) = \sqrt{1-x^2}$$

2) Hallar las siguientes integrales impropias:

$$a) \int_0^1 \frac{dx}{(x+x^2)^{\frac{1}{2}}}$$

$$b) \int_0^1 \frac{dx}{(x-x^2)^{\frac{1}{2}}}$$

$$c) \int_0^{\frac{\pi}{2}} \frac{x dx}{1-x^3}$$

$$d) \int_0^{\infty} \frac{dx}{(1+x^2)^2}$$

$$e) \int_0^1 \frac{\operatorname{sen} x}{x} dx$$

$$f) \int_0^1 \frac{e^x - 1}{x} dx$$

3) Representar gráficamente y hallar el área encerrada por las siguientes curvas:

$$a) y = x^3, y^2 = x$$

$$b) y = \frac{x^2}{4}, y = \frac{8}{x^2+4}$$

$$c) y = \frac{x}{2} + \operatorname{sen} x, y = 3x^{\frac{1}{4}}$$

4) En los siguientes ejercicios representar gráficamente y hallar:

(I) el volumen del sólido generado al rotar el arco en torno al eje x,

(II) la longitud de arco de la curva,

(III) el área de la superficie generada al rotar el arco en torno al eje x

$$a) y = \tan x; 1 \leq x \leq \frac{\pi}{3}$$

$$b) y = \frac{e^x + e^{-x}}{2}; 0 \leq x \leq 1$$